A two-port analogous circuit and SPICE model for Salmon's family of acoustic horns

W. Marshall Leach, Jr.
Georgia Institute of Technology, School of Electrical and Computer Engineering, Atlanta, Georgia 30332-0250

(Received 24 March 1995; accepted for publication 10 November 1995)

A two-port electroacoustic analogous circuit model for finite-length acoustic horns belonging to Salmon's family of horns is described. The circuit is useful for simulation of horn-loaded electroacoustic systems with electric circuit simulator software such as SPICE (simulation program with integrated circuit emphasis). An implementation of the circuit model for use with the *PSpice* simulator software is given. Example simulations are presented which illustrate applications of the model. © *1996 Acoustical Society of America.*

PACS numbers: 43.38.Ja, 43.20.Rz, 43.20.Wd

INTRODUCTION

Electroacoustic analogous circuits provide a powerful tool for the analysis of electro-mechano-acoustical systems. The power of these circuits becomes especially obvious when they are used with electric circuit simulator software such as SPICE.[1,2] The acronym SPICE stands for "simulation program with integrated circuit emphasis." When used with the personal computer, such software makes it possible to rapidly perform AC and transient analyses for complicated systems that would be difficult to perform otherwise.

SPICE is a public domain software developed at the University of California at Berkeley. Although the principal applications of the program are for the analysis of solid-state circuits, it can also be used to analyze linear passive networks. SPICE has been ported to the personal computer by several software vendors. The version which has been used in this paper is the evaluation version of *PSpice*, a product of the MicroSim Corp. The evaluation version is distributed free of charge by the vendor.

Historically, one of the more vexing problems in electroacoustic modeling has been the development of analogous circuits which represent the coupling between a transducer and the medium. In the case of a direct radiator, the concept of an imaginary, massless, rigid, freely movable piston across the transducer entry has been used to classically model the coupling. The impedance confronted by the piston is formulated in terms of Bessel functions. A major simplification was given in Ref. 3, where it was shown that a fixed resistor and inductor in parallel can be used to model the impedance. This representation and its variants are now commonly used when modeling radiation from a direct radiator.[4] An approximate solution to the problem of modeling the coupling when the transducer is used as a receiver was given in Ref. 5. A more complete solution was described in Ref. 6.

This paper concerns the development of an analogous circuit which models the coupling between a transducer and the medium when a finite-length acoustic horn separates the two. The horn is modeled by a two-port network in which voltage is analogous to pressure and current is analogous to volume velocity. It is assumed that the acoustic pressure in the horn satisfies the quasi-one-dimensional Webster horn equation[7] and that the horn is one of Salmon's family of horns.[8] A relevant application of such an analogous circuit is the simulation of electroacoustic systems with electric circuit simulator software. An implementation of the model for use with SPICE is given, and example simulations are presented which illustrate its application. The SPICE implementation of the model uses the analog behavioral modeling feature of *PSpice*.[9]

The methods presented in this paper require closed-form solutions for the pressure and volume velocity in a horn as a function of distance. For horn geometries in which the Webster equation cannot be solved, approximate numerical solutions can be obtained by approximating the horn by a cascade of short plane-wave tube sections. Such a solution is described in Ref. 10, where each section is modeled by a two-port network for which the elements in the transmission matrix are calculated. The overall solution is formulated as a product of the transmission matrices. Another approach is to model each plane-wave tube section with an analogous circuit consisting of a lumped-parameter transmission line. Such a solution is described in Ref. 11.

The functions representing acoustic and network variables in this paper are assumed to have time variations of the form $\exp(st)$, where s is the complex frequency. In the sinusoidal steady state, $s=j\omega=j2\pi f$, where ω is the angular frequency in rad/s and f is the frequency in Hz. Each variable is a phasor, with an amplitude and a phase. In the circuit diagrams, independent sources are represented by circular symbols. Dependent sources are represented by diamond-shaped symbols.

I. SALMON'S FAMILY OF HORNS

Figure 1 shows the schematic drawing of an acoustic horn of length L. Let the cross-sectional area of the horn as a function of the distance z into the horn be denoted by $A(z)$.



FIG. 1. Schematic drawing of acoustic horn of length L.

It is assumed that the acoustic pressure p is uniform across the cross section. In this case, the pressure satisfies the Webster horn equation[12] given by

$$\left\{ \frac{d^2}{dz^2} + \frac{1}{4A^2}\left[(A')^2 - 2AA''\right] - \frac{s^2}{c^2} \right\} A^{1/2} p = 0, \tag{1}$$

where s is the complex frequency, c is the velocity of sound, and the primes denote differentiation with respect to z. The corresponding acoustic volume velocity U satisfies Euler's equation

$$\rho_0 s U = -A\,\frac{dp}{dz}, \tag{2}$$

where ρ_0 is the density of air.

Solutions to the Webster equation are most easily obtained when

$$\frac{1}{4A^2}\left[(A')^2 - 2AA''\right] = -m^2, \tag{3}$$

where m is a constant. Horns which satisfy this condition belong to Salmon's family. The equation leads to solutions for the area $A(z)$ of the form

$$\begin{aligned} A(z) &= A_0\left[1 + \left(\frac{A_0'}{2A_0}\right) z\right]^2, \quad \text{for } m = 0 \\ &= A_0\left[\cosh mz + T \sinh mz\right]^2, \quad \text{for } m > 0, \end{aligned} \tag{4}$$

where $A_0 = A(0)$ is the area for $z = 0$, i.e., at the throat, and T is a constant. The case $m = 0$ describes a conical horn. The case $m > 0$ and $T = 0$ describes a catenoidal horn for which $A(z) = A_0 \cosh^2 mz$. The case $m > 0$ and $T = 1$ describes an exponential horn for which $A(z) = A_0 \exp(2mz)$. The case $m > 0$ and $0 < T < 1$ describes a hyperbolic horn.

For any member of Salmon's family of horns, the solutions for the acoustic pressure and volume velocity as functions of s and z can be written[12]

$$p(s,z) = \left[\frac{A_0}{A(z)}\right]^{1/2}\left[p_a e^{-\gamma(s)z} + p_b e^{+\gamma(s)z}\right], \tag{5}$$

$$\begin{aligned} U(s,z) = \left[\frac{A_0}{A(z)}\right]^{1/2} &\left[Y_a(s,z) p_a e^{-\gamma(s)z}\right. \\ &\left. - Y_b(s,z) p_b e^{+\gamma(s)z}\right], \end{aligned} \tag{6}$$

where p_a and p_b are constants having units of pressure and $\gamma(s)$, $Y_a(s,z)$, and $Y_b(s,z)$ are given by

$$\gamma(s) = \sqrt{(s/c)^2 + m^2}, \tag{7}$$

$$Y_a(s,z) = \frac{A(z)}{\rho_0 s}\left[\gamma(s) + \frac{A'(z)}{2A(z)}\right], \tag{8}$$

$$Y_b(s,z) = \frac{A(z)}{\rho_0 s}\left[\gamma(s) - \frac{A'(z)}{2A(z)}\right]. \tag{9}$$

These equations are similar to the equations for the voltage and current on an electrical transmission line. Unlike the transmission line, however, there are two characteristic admitances, $Y_a(s,z)$ for the forward propagating wave and $Y_b(s,z)$ for the reverse propagating wave. Both are functions of complex frequency s and distance z. In addition, the form of the propagation constant $\gamma(s)$ differs from that of a transmission line.

For $s = j\omega$, the propagation constant can be written

$$\begin{aligned} \gamma(j\omega) &= \sqrt{m^2 - (\omega/c)^2}, \quad \text{for } \omega/c < m, \\ &= j\sqrt{(\omega/c)^2 - m^2}, \quad \text{for } \omega/c > m. \end{aligned} \tag{10}$$

It follows that the term $\exp[-\gamma(s)L]$ represents an attenuation for $\omega/c < m$ and a time delay for $\omega/c > m$. The cutoff frequency of the horn is the frequency for which $\gamma = 0$. It is given by

$$f_c = \frac{\omega_c}{2\pi} = \frac{mc}{2\pi}. \tag{11}$$

II. AN ANALOGOUS CIRCUIT MODEL

Let the subscript 1 be used to denote the values of p, U, A, Y_a, and Y_b at $z = 0$. Similarly, let the subscript 2 be used to denote the values of p, U, A, Y_a, and Y_b at $z = L$. Equations (5) and (6) can be used to write the following:

$$p_1 = p_a + p_b, \tag{12}$$

$$U_1 = Y_{a1}(s) p_a - Y_{b1}(s) p_b, \tag{13}$$

$$p_2 = \left[\frac{A_1}{A_2}\right]^{1/2}\left[p_a e^{-\gamma(s)L} + p_b e^{+\gamma(s)L}\right], \tag{14}$$

$$U_2 = \left[\frac{A_1}{A_2}\right]^{1/2}\left[Y_{a2}(s) p_a e^{-\gamma(s)L} - Y_{b2}(s) p_b e^{+\gamma(s)L}\right]. \tag{15}$$

It is straightforward to use the above equations to show that p_1 and p_2 can be written in the following forms:

$$p_1 = U_1 Z_{11}(s) + T_1(s)\left[p_2 - U_2 Z_{12}(s)\right], \tag{16}$$

$$p_2 = T_2(s)\left[p_1 + U_1 Z_{21}(s)\right] - U_2 Z_{22}(s), \tag{17}$$



FIG. 2. Controlled-source analogous circuit for horn.

TABLE I. Definition of constants for example horns. ($k=10^3$, $m=10^{-3}$).

Constant	Definition	Horn A	Horn B
$k1$	$(A_1/A_2)^{1/2}$	10	40
$k2$	$(A_2/A_1)^{1/2}$	0.1	0.025
$k3$	$A_1'/2A_1$	1.821	1.821
$k4$	$A_2'/2A_2$	1.821	1.821
$k5$	$A_2'/2A_2-A_1'/2A_1$	0	0
$k6$	ρ_0/A_1	2329	2329
$k7$	ρ_0/A_2	23.29	1.455
$k8$	c	345	345
$k9$	c^2	119k	119k
$k10$	m	1.821	1.821
$k11$	m^2	3.317	3.317
$k12$	L	1.245	2.007
$k13$	L/c	3.608 m	5.816 m

where the $T_i(s)$ and $Z_{ij}(s)$ are transfer functions given by

$$T_1(s)=\left[\frac{A_2}{A_1}\right]^{1/2}\frac{\gamma(s)+A_2'/2A_2}{\gamma(s)+A_1'/2A_1}\,e^{-\gamma(s)L}, \tag{18}$$

$$Z_{11}(s)=\frac{\rho_0 s}{A_1[\gamma(s)+A_1'/2A_1]}, \tag{19}$$

$$Z_{12}(s)=\frac{\rho_0 s}{A_2[\gamma(s)+A_2'/2A_2]}, \tag{20}$$

$$T_2(s)=\left[\frac{A_1}{A_2}\right]^{1/2}\frac{\gamma(s)-A_1'/2A_1}{\gamma(s)-A_2'/2A_2}\,e^{-\gamma(s)L}, \tag{21}$$

$$Z_{21}(s)=\frac{\rho_0 s}{A_1[\gamma(s)-A_1'/2A_1]}, \tag{22}$$

$$Z_{22}(s)=\frac{\rho_0 s}{A_2[\gamma(s)-A_2'/2A_2]}. \tag{23}$$

Figure 2 shows an analogous circuit which models (16) and (17). It is clear that the circuit predicts the same equations for p_1 and p_2. For the special case of an exponential horn, the equations for $T_1(s)$ and $T_2(s)$ can be simplified. For the exponential horn, $A'/2A=2m$ so that $T_1(s)=(A_2/A_1)^{1/2}\exp[-\gamma(s)L]$ and $T_2(s)=(A_1/A_2)^{1/2}\times\exp[-\gamma(s)L]$.



FIG. 3. SPICE implementation of analogous circuit.

III. A SPICE MODEL FOR THE HORN

Figure 3 shows a SPICE subcircuit implementation of the horn analogous circuit with nodes labeled. The subcircuit code is given below. Constants which must be assigned numerical values are denoted by kj, where j is an integer. Table I defines these constants and gives their numerical values in mks units for the example simulations presented in the next section.

```
*HORN SUBCIRCUIT
.SUBCKT HORN 1 2
V1 1 3 OV
V2 5 2 OV
E11 3 4 LAPLACE {1(V1)}={k6*S/SQRT(S*S/k9+k11)+k3)}
E12 4 0 LAPLACE {V(13)}={1+k5/(SQRT(S*S/k9+k11)+k3)}
E13 7 8 2 0 1
E14 0 8 LAPLACE {I(V2)}={k7*S/(SQRT(S*S/k9+k11)+k4)}
T1 7 0 9 0 ZO=1K TD=k13
R11 9 0 1K
E15 13 0 LAPLACE {V(9)}={k2*EXP(k12*(S/k8−SQRT(S*S/k9+k11)))}
R12 13 0 1K
E22 0 6 LAPLACE {I(V2)}={k7*S/(SQRT(S*S/k9+k11)−k4)}
E21 5 6 LAPLACE {V(14)}={1+k5/(SQRT(S*S/k9+k11)−k4)}
E23 10 11 1 0 1
E24 11 0 LAPLACE {I(V1)}={k6*S/SQRT(S*S/k9+k11)−k3)}
T2 10 0 12 0 ZO=1K TD=k13
R21 12 0 1K
E25 14 0 LAPLACE {V(12)}={k1*EXP(k12*(S/k8−SQRT(S*S/k9+k11)))}
R22 14 0 1K
.ENDS HORN
```



FIG. 4. Circuit used to calculate throat acoustic impedance of example horns.

The lines in the code containing the word `LAPLACE` use the analog behavioral modeling feature of *PSpice*. On these lines, the circuit variable in braces to the left side of the equal sign is multiplied by the transfer function in braces to the right side of the equal sign. If any of these lines is continued, it must be broken so that the expressions in braces each fit on one line. The start of a continuation line must be indicated by a + sign followed by a space.

All sources in the circuit of Fig. 3 are voltage sources. The two sources `V1` and `V2` are used as ammeters to measure U_1 and U_2. The source `E11` implements the term $Z_{11}(s)U_1$. The source `E22` implements the term $Z_{22}(s)U_2$. The source `E14` implements the term $Z_{12}(s)U_2$. The voltage at node 7 is $p_2 - Z_{12}(s)U_2$. The source `E24` implements the term $Z_{21}(s)U_1$. The voltage at node 10 is $p_1 + Z_{11}(s)U_1$. The term $\exp[-\gamma(s)L]$ is implemented as the product of two terms as follows:

$$e^{-\gamma(s)L} = e^{-sL/c} \times e^{[s/c-\gamma(s)]L}, \tag{24}$$

where the term $\exp[-sL/c]$ represents a pure time delay that is implemented with a transmission line. Although $\exp[-\gamma(s)L]$ can be implemented directly, it was found that the product implementation defined by (24) significantly improved the accuracy of the computed phase of the term for frequencies well above the horn cutoff frequency.

Transmission lines `T1` and `T2` implement the pure time delay parts of the transfer functions $T_1(s)$ and $T_2(s)$. The sources `E12` and `E15` together implement $T_1(s)$ except for the pure time delay part. Similarly, sources `E21` and `E25` together implement $T_2(s)$ except for the pure time delay part. Resistors `R11` and `R21` are matched terminations for transmission lines `T1` and `T2`. Resistors `R12` and `R22` must be included to prevent nodes `13` and `14` from floating.

IV. NUMERICAL EXAMPLES

To illustrate applications of the horn model, the acoustic throat impedances of two exponential horns described in Ref. 13 are calculated in this section. In addition, the magnitude of the transfer function from the pressure at the throat to the on-axis radiated pressure in the far field is calculated for each horn. Both horns have a circular cross section. In each case, it is assumed that the mouth of the horn is terminated in an infinite baffle.

Figure 4 shows the circuit diagram for the calculations. The current source `IG` is analogous to a volume velocity source. It has an assigned AC current of 1 A in the code. It follows that the voltage at node 1 is equal to the acoustic throat impedance of the horn. Alternately, a voltage source could be used to drive the circuit. For a 1-V source, the acoustic throat impedance would be the reciprocal of the current in the source. The `V1` source is used as an ammeter to measure the output volume velocity. It has an assigned value of 0 V. The circuit elements `LMA1`, `RA1`, `RA2`, and `CA1` model the acoustic impedance of the infinite baffle air load. These element values are given by[4]

TABLE II. Numerical values for example horns ($\mu = 10^{-6}$, $m = 10^{-3}$, $k = 10^3$).

Quantity	Horn A	Horn B
Cutoff frequency	100 Hz	100 Hz
Throat diameter	1 in.	1 in.
Mouth diameter	10 in.	40 in.
Length	49 in.	79 in.
L_{MA1}	2.510	0.6276
R_{A1}	3543	221.5
R_{A2}	8034	502.1
C_{A1}	29.89μ	1.913 m
$\rho_0 c/A_1$	803.4k	803.4k

$$L_{MA1} = \frac{8\rho_0}{3\pi^2 a}, \tag{25}$$

$$R_{A1} = \frac{\rho_0 c}{\pi a^2}\left[\frac{128}{9\pi^2} - 1\right], \tag{26}$$

$$R_{A2} = \frac{\rho_0 c}{\pi a^2}, \tag{27}$$

$$C_{A1} = \frac{5.94 a^3}{\rho_0 c^2}, \tag{28}$$

where a is the radius of the mouth of the horn. The resistor `RL` in series with `LMA1` is necessary to prevent a loop with zero resistance. This resistor is assigned a value of 1 $\mu\Omega$. This is small enough to be considered a short circuit. The horn dimensions and values of the elements for each example are given in Table II. The last row in the table gives the high-frequency limit for the acoustic throat impedance.

The SPICE code for the example analyses is given below. Circuit labels for the air-load impedance elements have been used in place of numerical values. The horn subroutine described in the preceding section must be included at the indicated position.

```
EXAMPLE HORN CALCULATION
IG 0 1 AC 1A
X1 1 2 HORN
V1 2 3 AC 0V
LMA1 3 4 L_AM1
RL 4 0 1U
RA1 3 5 R_A1
RA2 5 0 R_A2
CA1 3 5 C_A1
(Code for Horn Subroutine Here)
.AC DEC 400 50 1.5K
.PROBE
.END
```



FIG. 5. (Upper) Real and imaginary parts of throat acoustic impedance for example horn A. (Lower) Magnitude of normalized pressure transfer function in dB for horn A.



FIG. 6. (Upper) Real and imaginary parts of throat acoustic impedance for example horn B. (Lower) Magnitude of normalized pressure transfer function in dB for horn B.

The upper plots in Figs. 5 and 6 show the calculated real and imaginary parts of the acoustic throat impedances as a function of frequency for the two example horns. In each case, the impedances are normalized by the high-frequency limit given in Table II. The real part is displayed by multiplying the magnitude by the cosine of the phase angle. The imaginary part is displayed by multiplying the magnitude by the sine of the phase angle. The figures show excellent agreement with those given in Ref. 13.

Let the on-axis far-field pressure radiated by the horn be denoted by p_3. For a horn with a circular cross section, this is given by

$$p_3 = \frac{\rho_0 s U_2}{2\pi r} e^{-sr/c}. \tag{29}$$

For $s = j2\pi f$, where f is the frequency in Hz, it follows that the magnitude of the pressure is given by

$$|p_3| = \frac{\rho_0 f |U_2|}{r}. \tag{30}$$

The lower plots in Figs. 5 and 6 show the calculated transfer function (in decibels) for $|p_3/p_1|$, where r is normalized to 1 m and p_1 is the pressure at the throat given by the voltage at node 1 in Fig. 4. The plots illustrate the usefulness of the model in calculating radiated pressure as a function of frequency. A more relevant application would replace the generator IG with the electroacoustic analogous circuit of a horn driver unit. In this case, the overall transfer function from the driver electrical input voltage to the on-axis pressure in the far field can easily be calculated.

Above the horn cutoff frequency, the plots for the pressure transfer functions both exhibit a positive aymptotic slope of $+20$ dB/dec. To obtain a zero asymptotic slope in this range, it follows that the plot of the throat pressure versus frequency must exhibit an asymptotic slope of -20 dB/dec. This must be achieved in the design of the horn driver unit.

V. CONCLUSIONS

Any finite-length acoustic horn in Salmon's family of horns can be modeled by a two-port analogous circuit in which the pressure at each port is expressed as a linear function of the pressure at the other port and the volume velocities flowing in both ports. The analogous circuit can be implemented in SPICE using the analog behavioral modeling feature of *PSpice* for personal computers. The SPICE model is useful for computer simulations of electroacoustic systems that employ horns. The methods used to derive the model for Salmon's family of horns are applicable to any family of horns for which the wave solutions for the pressure and volume velocity are known.

[1] W. M. Leach, Jr., "Computer-Aided Electroacoustic Design with SPICE," J. Audio Eng. Soc. **39**, 551–563 (July/Aug. 1991).

[2] W. M. Leach, Jr., "Controlled-Source Analogous Circuits and SPICE Models for Piezoelectric Transducers," IEEE Trans. Ultrasonics Ferroelec. Freq. Control **41**, 60–66 (Jan. 1994).

[3] B. B. Bauer, "Notes on Radiation Impedance," J. Acoust. Soc. Am. **15**, 223–224 (1944).

[4] L. L. Beranek, *Acoustics* (McGraw-Hill, New York, 1954).

[5] B. B. Bauer, "On the Equivalent Circuit of a Plane Wave Confronting an Acoustical Device," J. Acoust. Soc. Am. **42**, 1095–1097 (Nov. 1967).

[6] W. M. Leach, Jr., "On the Electroacoustic—Analogous Circuit for a Plane Wave Incident on the Diaphragm of a Free-Field Pressure Microphone," J. Audio Eng. Soc. **38**, 566–568 (July/Aug. 1990).

[7] A. G. Webster, "Acoustical Impedance and the Theory of Horns and of the Phonograph," Proc. Natl. Acad. Sci. (USA) **5**, 275–282 (1919).

[8] V. Salmon, "Generalized Plane Wave Horn Theory" and "A New Family of Horns," J. Acoust. Soc. Am. **17**, 199–211 and 212–218 (1946).

[9] *PSpice Circuit Analysis* (MicroSim Corp., Irvine, CA, 1990).

[10] J. S. McLean, J. T. Post, and E. L. Hixson, "A Theoretical and Experimental Investigation of the Throat Impedance Characteristics of Constant Directivity Horns," J. Acoust. Soc. Am. **92**, 2509–2520 (1992).

[11] B. N. Locanthi, "Application of Electric Circuit Analogies to Loudspeaker Design Problems," IRE Trans. Audio **PGA-6** (March 1952), and J. Audio Eng. Soc. **9**, 778–785 (Nov. 1971).

[12] A. D. Pierce, *Acoustics: An Introduction to Its Physical Principles and Applications* (McGraw-Hill, New York, 1981).

[13] H. F. Olson, *Acoustical Engineering* (Van Nostrand, Princeton, NJ, 1957).